

# Semantic GIS, GeoSPARQL

Michal Med

October 22, 2020


## 1 GraphDB

GeoSPARQL has full support in GraphDB.

### 1.1 Preparation

Download and import data into Graph DB – [https://graphdb.ontotext.com/documentation/9.0/free/\\_downloads/geosparql-example.rdf](https://graphdb.ontotext.com/documentation/9.0/free/_downloads/geosparql-example.rdf) and enable GeoSPARQL:

```
INSERT DATA {  
  _:s :enabled "true" .  
}
```



### 1.2 Spatial filters

Ex. 1 — Select all features that touches each other.

Answer (Ex. 1) —

```
PREFIX my: <http://example.org/ApplicationSchema#>
PREFIX geo: <http://www.opengis.net/ont/geosparql#>
PREFIX geof: <http://www.opengis.net/def/function/geosparql/>

SELECT ?f
WHERE {
  ?a my:hasExactGeometry ?aGeom .
  ?aGeom geo:asWKT ?aWKT .
  ?f my:hasExactGeometry ?fGeom .
  ?fGeom geo:asWKT ?fWKT .
  FILTER (geof:sfTouches(?aWKT, ?fWKT)
    && !sameTerm(?aGeom, ?fGeom))
}
```

Ex. 2 — Select all features overlapping rectangle D.

Answer (Ex. 2) —

```
PREFIX my: <http://example.org/ApplicationSchema#>
PREFIX geo: <http://www.opengis.net/ont/geosparql#>
PREFIX geof: <http://www.opengis.net/def/function/geosparql/>

SELECT ?f
WHERE {
  my:D my:hasExactGeometry ?dGeom .
  ?dGeom geo:asWKT ?dWKT .
  ?f my:hasExactGeometry ?fGeom .
  ?fGeom geo:asWKT ?fWKT .
  FILTER (geof:sfOverlaps(?dWKT, ?fWKT))
}
```

Ex. 3 — Select closest feature to the point.

```
"<http://www.opengis.net/def/crs/OGC/1.3/CRS84>
Point(-83.1 34.4)"^^geo:wktLiteral
```

Answer (Ex. 3) —

```
PREFIX uom: <http://www.opengis.net/def/uom/OGC/1.0/>
PREFIX my: <http://example.org/ApplicationSchema#>
PREFIX geo: <http://www.opengis.net/ont/geosparql#>
PREFIX geof: <http://www.opengis.net/def/function/geosparql/>
```

```

SELECT ?f ?dist
WHERE {
  ?f my:hasExactGeometry ?fGeom .
  ?fGeom geo:asWKT ?fWKT .
  BIND( (geof:distance(?fWKT, '''
    <http://www.opengis.net/def/crs/OGC/1.3/CRS84>
    Point(-83.1 34.4)'''^^geo:wktLiteral,
    uom:metre)) as ?dist)
} ORDER BY ASC(?dist) LIMIT 1

```

### 1.3 Spatial queries

Following exercises are for practising spatial queries

**Ex. 4** — Create feature AC as a union of features A and C.

**Answer (Ex. 4)** —

```

PREFIX my: <http://example.org/ApplicationSchema#>
PREFIX geo: <http://www.opengis.net/ont/geosparql#>
PREFIX geof: <http://www.opengis.net/def/function/geosparql/>

SELECT ?AC
WHERE {
  my:A my:hasExactGeometry ?aGeom .
  ?aGeom geo:asWKT ?aWKT .
  my:C my:hasExactGeometry ?cGeom .
  ?cGeom geo:asWKT ?cWKT .
  BIND( (geof:union(?aWKT, ?cWKT) as ?AC)
}

```

**Ex. 5** — Find all features closer than 10 kilometers to line E.

**Answer (Ex. 5)** —

```

PREFIX uom: <http://www.opengis.net/def/uom/OGC/1.0/>
PREFIX my: <http://example.org/ApplicationSchema#>
PREFIX geo: <http://www.opengis.net/ont/geosparql#>
PREFIX geof: <http://www.opengis.net/def/function/geosparql/>

SELECT ?f ?distance
WHERE {
  my:E my:hasExactGeometry ?eGeom .
  ?eGeom geo:asWKT ?eWKT .
  ?f my:hasExactGeometry ?fGeom .

```

```
?fGeom geo:asWKT ?fWKT .  
FILTER (?fGeom != ?eGeom)  
BIND((geof:distance(?eWKT, ?fWKT, uom:metre)) as ?distance)  
FILTER (?distance < 10000 && (?fGeom != ?eGeom))  
} ORDER BY ASC(?distance)
```